# $Id: PKGBUILD 182109 2013-04-07 09:14:17Z andyrtr $
# Maintainer: Andreas Radke <andyrtr@archlinux.org>

# Contributorr: Peter Lewis <plewis@aur.archlinux.org>
# Contributor: Sebastian Köhler <sebkoehler@whoami.org.uk>

# This must be built against the version of dovecot being used,
# else mail delivery will fail.
# Specify the version of dovecot to be used here:
_dcpkgver=2.1.16
# Make sure to bump pkgrel if changing this.

pkgname=pigeonhole
pkgver=0.3.4
pkgrel=1
pkgdesc="Fully rewritten Sieve implementation for Dovecot v2.1"
arch=('i686' 'x86_64')
url="http://pigeonhole.dovecot.org/"
license=('LGPL')
depends=('dovecot='$_dcpkgver)
conflicts=('dovecot-sieve' 'pigeonhole-hg')
source=("http://www.rename-it.nl/dovecot/2.1/dovecot-2.1-$pkgname-$pkgver.tar.gz"{,.sig}
        "dovecot.conf")
options=('!libtool')
sha256sums=('34da8801c7695c397a53a13574fe8674b92aa9d936f91a1b6e6df787488a6d05'
            'SKIP'
            'a457a1691cfa82495fc0503bfa4b61e54b149e63400fe0f568dff2c24a3f7858')

build() {
  cd "$srcdir/dovecot-2.1-$pkgname-$pkgver"

  ./configure --prefix=/usr \
    --with-dovecot=/usr/lib/dovecot \
    --disable-static
  make
}

package() {
  cd "$srcdir/dovecot-2.1-$pkgname-$pkgver"

  make DESTDIR="$pkgdir/" install
  install -m 644 -D "$srcdir/dovecot.conf" "$pkgdir/etc/ld.so.conf.d/dovecot.conf"
}
